Quickstart: *Mathematica* for Calculus I

C. G. Melles

Mathematics Department

United States Naval Academy

July 13, 2013

Contents

1. Getting started

Open a new *Mathematica* notebook by selecting File - New - Notebook. You will see a small white tab with a + sign at the top left. Click below and type a command, e.g., the first plot below, then press **shift-enter** to execute the command. You will see In[1]:= appear before your first line of input and Out[1]:= before your first output. Click below your output to enter another command. To enter some text, select **Format - Style - Text** or **Alt - 7**. To enter *Mathematica* commands again, click below the text and select **Format - Style - Input** or **Alt-9**. To delete a cell or cells, click on the vertical cell markers on the right of the screen and press **delete**.

2. Basic plotting

2.1. Plot the graph of sin (x) from x = 0 to x = 2 π.

Notice that Plot, Sin, and Pi must all be capitalized. Pay attention to the types of brackets used: [ ] vs { }.

```
Plot[Sin[x], {x, 0, 2 * Pi}]
```



2.2. Plot the graphs of the exponential function, the natural logarithm function, and the equation y=x on the same axes from x = -3 to x=3.

Note: In *Mathematica*, the natural logarithm function is Log. We use { } brackets for the list of expressions to be graphed.

```
Plot[{Exp[x], Log[x], x}, {x, -3, 3}]
```



2.3. Plot Exp[x], Log[x], and x from x=-3 to x=3 with the plot range of y restricted to [-3,3].

```
Plot[{Exp[x], Log[x], x}, {x, -3, 3}, PlotRange → {-3, 3}]
```



2.4. Plot the tangent and inverse tangent (arctangent) functions from $-\pi/2$ to $\pi/2$ with the first plot in red and the second in blue.

```
Plot[{Tan[x], ArcTan[x]}, {x, -Pi / 2, Pi / 2}, PlotStyle → {Red, Blue}]
```



2.5. Plot the top and bottom halves of the circle x^2 + y^2=1, adjusting the Aspect Ratio to show the same scales on both axes.

```
Plot[{-Sqrt[1 - x^2], Sqrt[1 - x^2]}, {x, -1, 1}, AspectRatio → Automatic]
```



3. Solving equations, approximating numerically, and finding roots

3.1. Solve the equation $x^4 - 4 = 0$ for x.

Note the double == signs in the *Mathematica* command. We see that two of the solutions are imaginary.

```
Solve[x^4 - 4 == 0, x]
```

$\{\{x \to -\sqrt{2}\}, \{x \to -i\sqrt{2}\}, \{x \to i\sqrt{2}\}, \{x \to \sqrt{2}\}\}$

3.2. Select the 4th solution from the list of solutions.

```
Solve[x^4 - 4 == 0, x][[4]]
```

$\{x \to \sqrt{2}\}$

3.3. Find the real solutions of the equation x^4-4=0.

```
Solve[x^4 - 4 == 0, x, Reals]
```

$\{\{x \to -\sqrt{2}\}, \{x \to \sqrt{2}\}\}$

3.4. Numerically approximate the real solutions of the equation x^4-4=0.

```
NSolve[x^4 - 4 == 0, x, Reals]
```

{{x → -1.41421}, {x → 1.41421}}

3.5. Find a solution of the equation x^4-4=0 in the interval 0<x<2.

```
Solve[x^4 - 4 == 0 && 0 < x < 2, x]
```

$\{\{x \to \sqrt{2}\}\}$

3.6. Numerically approximate a root of x^4-4=0 near x=1.5.

```
FindRoot[x^4 - 4, {x, 1.5}]
```

{x → 1.41421}

3.7. Numerically approximate the value of $\sqrt{2}$.

```
N[Sqrt[2]]
```

1.41421

4. Lists with the Table command

4.1. Use Table to make a list of the values of x^2 for integers x from 1 to 5.

Note that the output has { } brackets around it, indicating that it has the structure of a list.

```
Table[x^2, {x, 1, 5}]
```

{1, 4, 9, 16, 25}

4.2. Use Table to make a list of values of x^2 for x in {2,3,5,7,11,13,17} and name it PrimeSquares.

Note the { } brackets around the list of x values.

```
PrimeSquares = Table[x^2, {x, {2, 3, 5, 7, 11, 13, 17}}]
```

{4, 9, 25, 49, 121, 169, 289}

4.3. Select the 5th element of the list PrimeSquares.

```
PrimeSquares[[5]]
```

121

4.4. Construct a list of expressions c*x, for c in the list {-4,-2,-1,1,2,4}, and give it the name LineList.

```
LineList = Table[c * x, {c, {-4, -2, -1, 1, 2, 4}}]
```

{-4 x, -2 x, -x, x, 2 x, 4 x}

4.5. Use Table to plot the lines in the list LineList.

```
Plot[LineList, {x, -1, 1}]
```



4.6. Use Table to plot the lines in LineList with the colors orange, yellow, green,cyan,purple,black, and brown.

```
Plot[LineList, {x, -1, 1},
 PlotStyle → {Orange, Yellow, Green, Cyan, Purple, Black, Brown}]
```



5. Implicit plotting with ContourPlot

5.1. Plot the circle $x^2 + y^2$=1 using ContourPlot.

Note the double == signs in the command. We must specify the intervals for both x and y.

```
ContourPlot[x^2 + y^2 == 1, {x, -1, 1}, {y, -1, 1}]
```



5.2. Plot the same circle but with axes and no frame.

```
ContourPlot[x^2 + y^2 == 1, {x, -1, 1}, {y, -1, 1}, Axes → True, Frame → False]
```



5.3. Plot the family of circles $x^2 + y^2 = c^2$ for c in the list {1,2,4}, x in the interval [-4,4], and y in the interval [-4,4], using red, blue, and green.

Because of the order in which *Mathematica* executes commands, we need to tell the program to evaluate the list of equations before plotting with ContourPlot.

```
CircleEquations = Table[x^2 + y^2 == c^2, {c, {1, 2, 4}}]
```

$\{x^2 + y^2 == 1,\ x^2 + y^2 == 4,\ x^2 + y^2 == 16\}$

```
ContourPlot[Evaluate[CircleEquations],
 {x, -5, 5}, {y, -5, 5}, ContourStyle → {Red, Blue, Green}]
```



6. Defining and evaluating functions

6.1. Define $f(x) = x^3$.

```
f[x_] := x^3
```

6.2. Evaluate f at x=5 in two different ways. The arrow is obtained by typing - and >.

```
f[5]
```

125

```
f[x] /. x → 5
```

125

6.3. Define $g(x, y) = x^2 + y^2$.

```
g[x_, y_] := x^2 + y^2
```

6.4. Evaluate g at x=3, y=4 in two different ways.

```
g[3, 4]
```

25

```
g[x, y] /. {x → 3, y → 4}
```

25

7. Derivatives

7.1. Find the derivative of the function $f(x) = x^3$ defined above. Two methods are shown.

```
D[f[x], x]
```

$3 x^2$

```
f'[x]
```

$3 x^2$

7.2. Evaluate $f'(5)$ in two different ways.

```
f'[5]
```

75

```
D[f[x], x] /. x → 5
```

75

7.3. Find the second derivative of f. Two methods are shown.

```
f''[x]
```

6 x

```
D[f[x], {x, 2}]
```

6 x

7.4. Find the derivative of $x^2 + y^2$ with respect to x when y is a function of x. Note that we must specify that y is a function of x by writing y[x].

```
D[x^2 + y[x]^2, x]
```

2 x + 2 y[x] y′[x]

7.5. Given $x^2 + y^2$-25=0, solve for y'[x] by implicit differentiation. Give the result the name Slope.

```
Slope = Solve[D[x^2 + y[x]^2 - 25, x] == 0, y'[x]]
```

$\left\{\left\{y'[x] \to -\frac{x}{y[x]}\right\}\right\}$

7.6. Find the value of y'[x] at x=3, y=4. Note that we need to type y[x], not just y.

```
Slope /. {x → 3, y[x] → 4}
```

$\left\{\left\{y'[3] \to -\frac{3}{4}\right\}\right\}$

8. Combining graphs with the Show command

8.1. Plot the circle$x^2 + y^2$-25=0 and its tangent line at (3,4) on the same axes.

Note that we use ; at the end of a command when we do not need to see the result.

```
CircleGraph =
  ContourPlot[x^2 + y^2 - 25 == 0, {x, -6, 6}, {y, -6, 6}, Axes → True, Frame → False];
```

```
TangentGraph =
  Plot[-(3/4)*(x-3) + 4, {x, -6, 6}, PlotRange → {-6, 6}, AspectRatio → Automatic];
```

```
Show[CircleGraph, TangentGraph]
```



9. Symbols and Greek letters

9.1. The symbol ∞ is obtained from esc inf esc.

9.2. The symbol ≠ is obtained from esc ! = esc.

9.3. The symbol ≤ is obtained from esc < = esc (and similarly for ≥).

9.4. The Greek letter θ is obtained from esc theta esc. Other Greek letters are obtained in a similar manner.

10. The Simplify command

10.1. Simplify the expression $\cos^2(\theta) + \sin^2(\theta)$.

```
Simplify[Cos[θ]^2 + Sin[θ]^2]
```

```
1
```

11. Limits

11.1. Find the limit of Exp[-x] as x→∞.

```
Limit[Exp[-x], x → Infinity]
```

```
0
```

11.2. Find the limit of Log[x] (natural logarithm) as $x \to 0^+$.

```
Limit[Log[x], x → 0, Direction → -1]
```

```
-∞
```

11.3. Find the limit of 1/x as $x \to 0^-$.

```
Limit[1 / x, x → 0, Direction → 1]
```

```
-∞
```

11.4. Find the limit of Exp[-c*x] as x→∞, with the assumption that c>0.

```
Limit[Exp[-c * x], x → Infinity, Assumptions → {c > 0}]
```

0

11.5. Find the limit of Exp$\left[-c^2\right.$*x] as x→∞, with the assumption that c is a real number and c≠0. Note: Type != or esc != esc for ≠.

```
Limit[Exp[-c^2 * x], x → Infinity, Assumptions → Element[c, Reals] && c ≠ 0]
```

0

12. Integrals

12.1. Find the indefinite integral of x^2. *Mathematica* does not supply the +C.

```
Integrate[x^2, x]
```

$\frac{x^3}{3}$

12.2. Find the definite integral of x^2 from 0 to 1.

```
Integrate[x^2, {x, 0, 1}]
```

$\frac{1}{3}$

12.3. Numerically approximate the integral of x^2 from 0 to 1.

```
NIntegrate[x^2, {x, 0, 1}]
```

0.333333